Mandl and Shaw

pages 113 - 115; 125 - 126; 173

THE MØLLER CROSS SECTION

This is the cross section for *electron-electron elastic scattering*.

M_t



$$(\bar{u}_3\, ie\gamma^\mu u_1)(\bar{u}_4\, ie\gamma^\nu u_2)\frac{-ig_{\mu\nu}}{t}$$

$$\frac{e^2}{t}(\bar{u}_3\gamma^\mu u_1)(\bar{u}_4\gamma^\nu u_2)\, g_{\mu\nu}$$

M_u



$$k^\mu = p_1^\mu - p_4^\mu = p_3^\mu - p_2^\mu$$

$$\frac{-e^2}{u}(\bar{u}_4\gamma^\mu u_1)(\bar{u}_3\gamma^\nu u_2)\, g_{\mu\nu}$$

The matrix element is $M = M_t + M_u$.

The cross section requires $|M|^2$,

$$= |M_t|^2 + |M_u|^2 + M_t M_u^* + M_t^* M_u .$$

$|M_t|^2$

$$\mathcal{M}_t = \frac{e^2}{t}(\bar{u}_3\gamma^\mu u_1)(\bar{u}_4\gamma^\nu u_2)\, g_{\mu\nu}$$

$$\mathcal{M}_t^* = \frac{e^2}{t}(\bar{u}_3\gamma^\mu u_1)^*(\bar{u}_4\gamma^\nu u_2)^*\, g_{\mu\nu}$$

$$(\bar{u}_3\gamma^\mu u_1)^* = (u_3^\dagger\gamma^0\gamma^\mu u_1)^*$$

$$= u_1^\dagger(\gamma^\mu)^\dagger(\gamma^0)^\dagger u_3$$

Properties of γ^μ: $(\gamma^\mu)^\dagger = \gamma^0\gamma^\mu\gamma^0$;
(EQ. 4.16)

$(\gamma^0)^\dagger = \gamma^0\gamma^0\gamma^0 = \gamma^0$; and $(\gamma^0)^2 = 1$.

$$\hookrightarrow = \bar{u}_1\,\gamma^0(\gamma^\mu)^\dagger\gamma^0\,u_3 = \bar{u}_1\gamma^\mu u_3$$

So

$$\mathcal{M}_t^* = \frac{e^2}{t}(\bar{u}_1\gamma^\alpha u_3)(\bar{u}_2\gamma^\beta u_4)\, g_{\alpha\beta}$$

$$|\mathcal{M}_t|^2 = \frac{e^4}{t^2}(\bar{u}_3\gamma^\mu u_1)(\bar{u}_1\gamma^\alpha u_3)(\bar{u}_4\gamma^\nu u_2)(\bar{u}_2\gamma^\beta u_4)\, g_{\mu\nu}\cdot g_{\alpha\beta}$$

$$= \frac{e^4}{t^2}(\bar{u}_3\gamma^\mu u_1)(\bar{u}_1\gamma^\alpha u_3)(\bar{u}_4\gamma_\mu u_2)(\bar{u}_2\gamma_\alpha u_4)$$

$$|\mathcal{M}_t|^2 = \frac{e^4}{t^2} \mathrm{Tr}\{ \gamma^\mu u_1 \bar{u}_1 \gamma^\alpha u_3 \bar{u}_3 \} \mathrm{Tr}\{ \gamma_\mu u_2 \bar{u}_2 \gamma_\alpha u_4 \bar{u}_4 \}$$

For unpolarized scattering,

average over r_1 and r_2 $\Rightarrow \frac{1}{4} \sum_{r_1=1}^{2} \sum_{r_2=1}^{2}$

and sum over r_3 and r_4 $\Rightarrow \sum_{r_3=1}^{2} \sum_{r_4=1}^{2}$

Recall, $\sum_{r=1}^{2} u(\vec{p}, r)\bar{u}(\vec{p}, r) = \frac{\not{p}+m}{2m} \equiv \Lambda^+(p)$

$$\overline{|\mathcal{M}_t|^2} = \frac{e^4}{4t^2} \frac{1}{(2m)^4} L_{13}^{\mu\alpha} L_{24\,\mu\alpha}$$

$$L_{13}^{\mu\alpha} = \mathrm{Tr}\{ \gamma^\mu (\not{p}_1 + m) \gamma^\alpha (\not{p}_3 + m) \}$$

$$L_{24}^{\mu\alpha} = \mathrm{Tr}\{ \gamma^\mu (\not{p}_2 + m) \gamma^\alpha (\not{p}_4 + m) \}$$

Now use FeynCalc.

Moller ee -> ee

In[1]:=
```
$LoadFeynArts = False;
<< HighEnergyPhysics`FeynCalc`
```

Loading FeynCalc from /home/stump/.Mathematica/Appl

FeynCalc 8.2.0 For help, type ?FeynCalc, open FeynCa

In[3]:=
```
(* scalar products *)
ScalarProduct[p1, p1] = m^2;
ScalarProduct[p2, p2] = m^2;
ScalarProduct[p3, p3] = m^2;
ScalarProduct[p4, p4] = m^2;
(* Mandelstam variables *)
ScalarProduct[p1, p2] = (s - 2 m^2) / 2;
ScalarProduct[p3, p4] = (s - 2 m^2) / 2;
ScalarProduct[p1, p3] = (2 m^2 - t) / 2;
ScalarProduct[p2, p4] = (2 m^2 - t) / 2;
ScalarProduct[p1, p4] = (2 m^2 - u) / 2;
ScalarProduct[p2, p3] = (2 m^2 - u) / 2;
```

|M_t|^2

In[70]:=
```
frontTT = e^4 / (4 * t^2) / (2 m)^4;
L13 = Tr[GA[μ].(GS[p1] + m).GA[α].(GS[p3] + m)];
L24 = Tr[GA[μ].(GS[p2] + m).GA[α].(GS[p4] + m)];
ξTT = Contract[L13.L24];
ξTT = Simplify[ξTT]
ξTT = Expand[ξTT /. {m^2 → (s + t + u) / 4}]
```

Out[74]= $8\left(8\, m^4 - 4\, m^2\,(s - t + u) + s^2 + u^2\right)$

Out[75]= $64\, m^4 - 16\, s\, u + 8\, t^2$

|M_u|^2

In[76]:=
```
frontUU = e^4 / (4 * u^2) / (2 m)^4;
L14 = Tr[GA[μ].(GS[p1] + m).GA[α].(GS[p4] + m)];
L23 = Tr[GA[μ].(GS[p2] + m).GA[α].(GS[p3] + m)];
ξUU = Contract[L14.L23];
ξUU = Simplify[ξUU]
ξUU = Expand[ξUU /. {m^2 → (s + t + u) / 4}]
```

Out[80]= $8\left(8\, m^4 - 4\, m^2\,(s + t - u) + s^2 + t^2\right)$

Out[81]= $64\, m^4 - 16\, s\, t + 8\, u^2$

Comment:

Note that $M_u^{\,2}$ is equal to $M_t^{\,2}$ with the substitutions $t \rightleftarrows u$, i.e., $p_3^{\mu} \rightleftarrows p_4^{\mu}$.

The interference terms,

$$M_t\, M_u^* + M_t^*\, M_u \,.$$

$$\mathcal{M}_t = \frac{e^2}{t}\,(\bar{u}_3\,\gamma^\mu\,u_1)(\bar{u}_4\,\gamma^\nu u_2)\,g_{\mu\nu}$$

$$\mathcal{M}_u = \frac{-e^2}{u}\,(\bar{u}_4\,\gamma^\mu\,u_1)(\bar{u}_3\,\gamma^\nu\,u_2)\,g_{\mu\nu}$$

$$\mathcal{M}_u^* = -\frac{e^2}{u}\,(\bar{u}_1\,\gamma^\alpha\,u_4)(\bar{u}_2\,\gamma^\beta\,u_3)\quad g_{\alpha\beta}$$

$$\mathcal{M}_t\,\mathcal{M}_u^* = \frac{-e^4}{tu}\,(\bar{u}_3\gamma^\mu u_1)(\bar{u}_4\gamma^\nu u_2)\quad g_{\mu\nu}$$
$$(\bar{u}_1\gamma^\alpha u_4)(\bar{u}_2\gamma^\beta u_3)\quad g_{\alpha\beta}$$

$$= \frac{-e^4}{tu}\,(\bar{u}_3\gamma^\mu u_1)(\bar{u}_1\gamma^\alpha u_4)(\bar{u}_4\,\gamma^\nu u_2)(\bar{u}_2\gamma^\beta u_3)$$
$$g_{\mu\nu}\,g_{\alpha\beta}$$

$$\overline{\mathcal{M}_t\mathcal{M}_u^*} = \frac{-e^4}{4tu}\,\frac{1}{(2m)^4}\,g_{\mu\nu}\,g_{\alpha\beta}$$
$$\mathrm{Tr}\left\{\gamma^\mu(\not{p}_1+m)\gamma^\alpha(\not{p}_4+m)\,\gamma^\nu(\not{p}_2+m)\gamma^\beta(\not{p}_3+m)\right\}$$

requires trace of product of 8 matrices.

Then contract with $g_{\mu\nu}\,g_{\alpha\beta}$.

Then substitute $p_i \cdot p_j$ with Mandelstam variables.

Feyncalc.

Interference terms

```
In[55]:= frontINT = -e^4 / (4 * t * u) / (2 m) ^4;
         ξINT = Tr[GA[μ].(GS[p1] + m).GA[α].(GS[p4] + m).
              GA[μ].(GS[p2] + m).GA[α].(GS[p3] + m)];
         ξINT = Simplify[Expand[ξINT]];
         ξINT = ξINT /. {u → 4 * m^2 - s - t};
         ξINT = Simplify[Expand[ξINT]]
         compare = Expand[(s - 2 * m^2) * (s - 6 * m^2)]
         ξINT = -8 * compare
```

Out[59]= $-8\left(12\, m^4 - 8\, m^2\, s + s^2\right)$

Out[60]= $12\, m^4 - 8\, m^2\, s + s^2$

Out[61]= $-8\left(12\, m^4 - 8\, m^2\, s + s^2\right)$

So, the matrix element squared, for unpolarized scattering, is

Matrix element squared

```
In[163]:= MTT = Simplify[frontTT * ξTT];
          MUU = Simplify[frontUU * ξUU];
          MINT = Simplify[2 * frontINT * ξINT ];
          Msq = MTT + MUU + MINT
```

Out[166]= $$\frac{e^4\left(8\, m^4 - 2\, s\, u + t^2\right)}{8\, m^4\, t^2} + \frac{e^4\left(8\, m^4 - 2\, s\, t + u^2\right)}{8\, m^4\, u^2} + \frac{e^4\left(2\, m^2 - s\right)\left(6\, m^2 - s\right)}{4\, m^4\, t\, u}$$

The cross section

Let's calculate the cross section in the center of mass frame of reference.

$$d\sigma = \frac{\overbrace{(2m)^4}^{\equiv (2m_a)(2m_b)(2m_c)(2m_d)}}{|v_1-v_2|\,4E_1E_2}\int (2\pi)^4\delta^4(P_f-P_i)\int \frac{d^3p_3}{(2\pi)^3\, 2E_3}\,\frac{d^3p_4}{(2\pi)^3\, 2E_4}\,|\mathcal{M}|^2$$

$$= \frac{(2m)^4}{|v_1-v_2|\,4E_1E_2}\,\frac{1}{4\pi^2}\int \delta(E_f-E_i)\,\frac{p_3^2\,dp_3}{4E_3E_4}\,|\mathcal{M}|^2$$

$$\frac{d\sigma}{d\Omega_3} = \frac{(2m)^4}{|v_1-v_2|\,E_1E_2}\,\frac{1}{64\pi^2}\int \delta(E_f-E_i)\,\frac{p_3^2\,dp_3}{E_3E_4}\,|\mathcal{M}|^2$$

$$\frac{d\sigma}{d\Omega_3} = \frac{(2m)^4}{|v_1-v_2|\,E_1E_2}\,\frac{1}{64\pi^2}\int \delta(E_f-E_i)\,\frac{p_3^2\,dp_3}{E_3E_4}\,|\mathcal{M}|^2$$

Consider $a+b\to c+d$ in the C.o.M. frame of reference



Calculate $C_1 = \int_0^\infty \delta(E_3+E_4-E_1-E_2)\,\frac{p_3^2\,dp_3}{E_3E_4}$

$$C_1 = \int_0^\infty \delta[F(p_3)]\,\frac{p_3^2\,dp_3}{E_3E_4} = \frac{1}{[dF/dp_3]}\,\frac{p_3^2}{E_3E_4}$$

$$F = \sqrt{p_3^2+m_c^2}+\sqrt{p_3^2+m_d^2}-\sqrt{s} \qquad (\text{Note: } E_1+E_2=\sqrt{s})$$

$$\frac{dF}{dp_3} = \frac{p_3}{E_3}+\frac{p_3}{E_4} = \frac{p_3(E_3+E_4)}{E_3E_4} = \frac{p_3\sqrt{s}}{E_3E_4} \Rightarrow C_1 = \frac{p_3}{\sqrt{s}}$$

Calculate $C_2 = |v_1-v_2|\,E_1E_2$

$$C_2 = \left|\frac{p_1}{E_1}+\frac{p_1}{E_2}\right| E_1E_2 = p_1(E_1+E_2) = p_1\sqrt{s}$$

$$\frac{d\sigma}{d\Omega_3} = \frac{(2m)^4}{p_1\sqrt{s}}\,\frac{1}{64\pi^2}\,\frac{p_3}{\sqrt{s}}\,|\mathcal{M}|^2$$

$$\frac{d\sigma}{d\Omega_3} = \frac{(2m)^4}{64\pi^2 s}\,\frac{p_3}{p_1}\,|\mathcal{M}|^2 \qquad \text{where } (2m)^4 = \prod_{i=1}^{4}(2m_i)$$

For Møller scattering ($ee \to ee$)

$m_a = m_b = m_c = m_d = m$;

in the C.o.M. frame,

$E_1 = E_2 = E_3 = E_4$ and $|\vec{p}_3| = |\vec{p}_1|$;

$$\frac{d\sigma}{d\Omega_3} = \frac{(2m)^4}{64\pi^2 s}|\mathcal{M}|^2$$

<u>The center of mass frame of reference</u>

I.e., $\mathbf{P_i} = \mathbf{P_f} = 0$. Picture:



All masses $= m$.

$\therefore E_2 = E_1$

$\cdot\ E_3 = E_4$

$E_1 + E_2 = E_3 + E_4$

$E_i = E_1$ all i

$p_i = p_1$ all i

$$s = (p_1 + p_2)^2 = (E_1 + E_2)^2 = 4E_1^2$$

$$\begin{aligned}
t = (p_1 - p_3)^2 &= m^2 + m^2 - 2p_1 \cdot p_3 \\
&= 2m^2 - 2E_1E_3 + 2\vec{p}_1 \cdot \vec{p}_3 \\
&= 2m^2 - 2E_1^2 + 2p_1p_3 \cos\theta \\
&= -2p_1^2 + 2p_1p_3 \cos\theta \\
&= -2p_1^2(1 - \cos\theta) = -2(E_1^2 - m^2)(1 - \cos\theta)
\end{aligned}$$

$$u = -2(E_1^2 - m^2)(1 + \cos\theta)$$

Plot the CoM cross section; using units of MeV and mb;

In[35]:=
```
dσ = (2 * m) ^4 / (64 * Pi ^2 * s) * Msq;
dσ = dσ /. {e → Sqrt[4 * Pi * α]};
temp = Expand[dσ / α ^2]
const = (197) ^2 * 1000 / 100 (* hbc=197 MeV.fm ; 1 barn = 100 fm^2 *)
dσ = dσ * const;
```

Out[37]=
$$\frac{4\,m^4}{s\,t^2}+\frac{12\,m^4}{s\,t\,u}+\frac{4\,m^4}{s\,u^2}-\frac{8\,m^2}{t\,u}+\frac{s}{t\,u}+\frac{1}{s}-\frac{u}{t^2}-\frac{t}{u^2}$$

Out[38]= 388 090

In[58]:=
```
(* numerical *)
CS = dσ /. {α → 1 / 137};
CS = CS /.  {s -> 4 E1 ^2,
      t -> -2 * (E1 ^2 - m ^2) * (1 - cos),
      u -> -2 * (E1 ^2 - m ^2) * (1 + cos)};
CS = CS /. {m → 0.511} ;
CS = CS /. {E1 → 1.30}; (* E1 → electron energy in MeV *)
CS = Simplify[Expand[CS]]
```

Out[61]=
$$\frac{29.8668+3.05875\cos^4+25.3667\cos^2}{1.+\cos^4-2.\cos^2}$$

In[62]:=
```
tbl = {}
Do[ th = 0.01 * i;
 CrSec = CS /. {cos → Cos[th]};
 tbl = Join[tbl, {{th, CrSec}}],
 {i, 1, 314}]
tbl;
ListLogPlot[tbl, PlotRange → {{0, 3.2}, {1, 1*^6}},
 Frame → True,
 FrameLabel → {"θ/π", "dσ/dΩ in mb"},
 PlotLabel → "Moller cross section for E1 = 1.30 MeV"]
```

Out[62]= {}

Out[65]=



Homework Problems due Friday March 31 **10.** Mandl and Shaw, Problem 8.6. Plot a graph of the cross section in the high-energy limit.